
Modelos administrativos
en Sheets
Martes 19 de octubre de 2027. Rosalba, de Estudio Roble, pregunta

cuánto cambiaría el presupuesto PR-001 (138,16 €) con 11 unidades

de A001 en vez de 10, y Aldara quiere saber qué sigue pendiente

tras cancelarse el pedido P004. Carolina teclea los precios dentro

de las fórmulas. ¿Cómo se construye el modelo que debió existir

desde septiembre?

JoseSande.com LIBROS  DE  TEXTO  GRAT IS



PARA  EMPEZAR

En esta unidad aprenderás a…

🎯
Importar un archivo de datos a una hoja de

cálculo sin perder el original y dejar

constancia de cómo se importó.
🧭 Normalizar tipos (texto, número y fecha) para

que las fórmulas operen sobre datos fiables.

🔍
Construir un modelo con entradas, cálculos y

resultados separados, con los supuestos fuera

de las fórmulas.
💡

Relacionar tarifas con búsquedas y hacer

visible una tarifa ausente en lugar de

sustituirla por cero.

🛠️
Calcular vencimientos y cobros, validar

entradas, alterar un supuesto y comprobar los

totales por un camino independiente.

S ITUACIÓN  DE  PART IDA

«Los supuestos fuera de la fórmula»

❓ ¿Qué datos del presupuesto PR-001 son
entradas, cuáles son cálculos y cuáles son
resultados?

❓ Si el precio de un artículo está escrito
dentro de cien fórmulas, ¿qué ocurre el día
que cambia?

❓ ¿Qué debería hacer una hoja cuando no
encuentra la tarifa de un artículo: poner un
cero, dejar el hueco o avisar?

❓ Una factura emitida el 04/10/2027, ¿es
dinero cobrado? ¿Qué dato falta para
saberlo?

Martes 19 de octubre de 2027. Rosalba, de

Estudio Roble, pasa por Nexo Oficina (empresa

ficticia) y pregunta a Carolina cuánto cambiaría

el presupuesto PR-001, hecho a mano el jueves

16/09/2027 para el pedido P001 (total de 138,16

€), si en vez de 10 unidades de A001 fueran 11.

Además, ayer lunes se canceló el pedido P004 y

Aldara quiere saber qué sigue pendiente.

Carolina abre una hoja y, para ir rápido, teclea los

precios dentro de las fórmulas: =C2*3,5. Teodoro

mira la pantalla y dice: «Los supuestos fuera de la

fórmula».

MAPA  DE  LA  UNIDAD

Lo que vamos a ver

6.1 Importar datos 6.2 Normalizar tipos 6.3 Construir modelo

6.4 Relacionar tarifas 6.5

Calcular cobros; introducir
validaciones; alterar un supuesto;
comprobar totales de forma
independiente

6 . 3  ·  ONCE  UNIDADES

Cambio una celda y se recalcula todo

🛒

📊
🔎

CLAVE

Un modelo reproduce una decisión con datos que puedes cambiar: si los supuestos están en celdas, basta

cambiar C2 para responder a Rosalba; si una tarifa falta, el modelo tiene que decirlo.

Si en vez de diez fueran once de A001, ¿cuánto cambia?

Cambio una celda y se recalcula todo el presupuesto.

Muy bien. ¿Y si falta una tarifa?

6 .1  ·  IMPORTAR  DATOS

Del archivo a la hoja

📄
El CSV

Una línea por registro y valores separados por una coma o

un punto y coma. Importarlo evita reteclear, donde se

cuelan los errores.

⚙️
Archivo › Importar › Subir

Insertar nuevas hojas, separador Personalizado (punto y

coma) y conversión del texto en números, fechas y fórmulas

desactivada. La hoja se llama Origen_tarifas.

🔤
Todo llega como texto

Con la conversión desactivada lo conviertes tú, con una

regla escrita; con ella activada, Sheets decidiría por su

cuenta. Región: España (Archivo › Configuración).

🔗
Importar no es enlazar

Importar hace una copia estática. IMPORTRANGE("dirección

de la hoja";"Tarifas!A1:C7") crea un vínculo vivo: se actualiza,

pero exige el acceso de su propietario.

CLAVE

El original no se toca: Origen_tarifas queda tal como llegó y lo corregido va a otra hoja, con fórmulas o pasos anotados.

6 .1  ·  DOCUMENTO  1

Exportación de tarifas con tres defectos
2027-10-19_tarifas_nexo_v01.csv, exportado por el programa de gestión de Nexo Oficina, con

punto y coma como separador (ficticio).

CÓDIGO DESCRIPCIÓN PRECIO VIGENTE DESDE

A001 Carpeta azul de recambio 3,50 2027-09-01

A002␣ Bandeja ámbar de recambio 12.80 2027-09-01

A003 Cajón verde de recambio 11.33 2027-09-01

A004 Portalápices ámbar 4,50 2027-09-01

A005 Vade de escritorio ámbar 22,00 2027-09-01

A006 Separadores de archivo, pack 10,00 2027-09-01

Comprueba: siete filas y cuatro columnas; =CONTARA(A2:A) devuelve 6 y =CONTAR(C2:C) devuelve 0: ningún precio es aún un número, ni el de 3,50.

CLAVE

Registro de importación: fuente y fecha (19/10/2027 · Carolina), filas esperadas e importadas (6 y 6), separador y conversión

(punto y coma, desactivada) y defectos. Regla de Teodoro: «Fuente, paso y repetición».

6 . 2  ·  NORMALIZAR  T IPOS

Una regla por defecto

Hoja Tarifas · fórmulas de la fila 3 (A002), copiadas desde la
fila 2

Espacio sobrante en el código =ESPACIOS(Origen_tarifas!A3) A002

Punto decimal en el precio =VALOR(SUSTITUIR(Origen_tarifas!C3;".";",")) 12,8

Fecha año-mes-día =FECHA(EXTRAE(Origen_tarifas!D3;1;4);EXTRAE(…;6;2);EXTRAE(…;9;2)) 01/09/2027

Toda la columna de una vez =ARRAYFORMULA(VALOR(SUSTITUIR(Origen_tarifas!C2:C7;".";","))) de 3,5 a 10

Pruebas: =ESNUMERO(C2) VERDADERO, =CONTAR(C2:C7) = 6 y =LARGO(A3) = 4 (antes, 5); después, Pegado especial › Solo valores, anotado. Las fechas son
días desde el 30/12/1899 (04/10/2027 = 46664) y una celda vacía no es un cero: 3,50 · 12,80 · vacío · 4,50 promedian 6,93; con un 0, 5,20.

CLAVE

Recortar espacios en blanco actúa en el sitio y no deja rastro; la fórmula sí. Nunca se teclea encima del dato.

6 . 3  ·  CONSTRU IR  MODELO

Entradas, cálculos y resultados

✍️
Entradas

Lo que alguien decide o teclea: cantidades, descuento, IVA,

plazo. Relleno suave y etiqueta con su unidad: el color solo

no basta en blanco y negro.

⚙️
Cálculos y resultados

Fórmulas que no se tocan, y lo que se enseña: total, estado,

alertas. Si se mezclan, no se sabe qué celda se puede

cambiar.

📌
Supuestos fuera de la fórmula

=C2*3,5 o =E6*0,05 esconden el dato. Bien:

=REDONDEAR(C2*D2;2) y =REDONDEAR(E6*$H$2;2), con el

supuesto en su celda.

🏷️
Intervalos con nombre

Datos › Intervalos con nombre: DESCUENTO, IVA y

PLAZO_COBRO para H2, H3 y H4.

=REDONDEAR(E6*DESCUENTO;2) se lee como una frase.

CLAVE

Presupuesto: cabeceras en A1:E1, líneas de PR-001 en A2:C4, supuestos en G2:H4, cálculos en D6:E9 y resultado en D10:E10.

Revisa con Ver › Mostrar › Fórmulas: una constante dentro de una fórmula es un defecto.

6 . 3  ·  PR IMERA  VERS IÓN

PR-001 como modelo
Precios enlazados en D2 con =Tarifas!C2, copiada a D3 y D4. Supuestos: descuento (H2) 5 %,

IVA (H3) 21 %.

CELDA FÓRMULA RESULTADO (€)

E2:E4 =REDONDEAR(C2*D2;2) (copiada) 35,00 · 51,20 · 33,99

E6 =SUMA(E2:E4) 120,19

E7 =REDONDEAR(E6*$H$2;2) 6,01

E8 =E6-E7 114,18

E9 =REDONDEAR(E8*$H$3;2) 23,98

E10 =SUMA(E8:E9) 138,16

E10 muestra 138,16, el total a mano del 16/09/2027. Líneas: A001 × 10, A002 × 4 y A003 × 3.

CLAVE

Si se ordenara Tarifas, el enlace =Tarifas!C2 se rompería: lo resuelve BUSCARV (6.4).

🙋  PREGUNTA  PARA  LA  CLASE

Rosalba pregunta por 11 unidades de A001. Carolina
cambia solo C2. ¿Qué total muestra E10?

✔ Subtotal 123,69; descuento 6,18; base 117,51; IVA 24,68; total 142,19 €: 4,03 € más que con 10 unidades
(actividad 11).

A 142,18 €

B 142,19 €

C 138,16 €

6 .4  ·  RELACIONAR  TAR IFAS

Una búsqueda por el código

=BUSCARV(A2;Tarifas!$A$2:$C$7;3;FALSO)

Clave A2
el código de la

línea

Rango Tarifas!$A$2:$C$7
la tabla, anclada

con $

Columna 3 el precio

FALSO coincidencia exacta

sin él, puede
traer el precio de

otro artículo

D2 devuelve 3,5; D3, 12,8; D4, 11,33. Ya puedes ordenar Tarifas: el precio sigue a su código. Si el precio estuviera a la izquierda:
=INDICE(Tarifas!$C$2:$C$7;COINCIDIR(A4;Tarifas!$A$2:$A$7;0)) devuelve 11,33 para A003.

CLAVE

El enlace =Tarifas!C2 sigue a la celda; la búsqueda sigue al código.

6 .4  ·  CUANDO  FALTA  LA  TAR IFA

Tres reacciones, solo una buena
En una copia, sin la fila de A003 en Tarifas, BUSCARV devuelve #N/A (dato no disponible; en

Calc, #N/D) y el error se propaga a E4, E6 y el total.

REACCIÓN FÓRMULA DE D4 TOTAL (€) VALORACIÓN

Dejar que falle =BUSCARV(A4;Tarifas!$A$2:$C$7;3;FALSO) #N/A Visible, pero poco explicativo

Ocultar con cero =SI.ERROR(BUSCARV(…);0) 99,09 Peligroso: parece correcto

Detectar y avisar Columna F de aviso #N/A y aviso Correcto

El error silencioso: 99,09 € en lugar de 138,16 € (39,07 € menos) y nadie lo nota. SI.ERROR sirve cuando la alternativa es honesta («sin dato»); daña
cuando convierte un fallo en un cero que se suma.

CLAVE

Un hueco no es un cero: el modelo tiene que fallar a la vista o avisar con un texto claro.

6 .4  ·  AV ISO  E  INCIDENCIA

FALTA TARIFA y su registro
F2 (copiada a F3:F4): =SI(ESNOD(COINCIDIR(A2;Tarifas!$A$2:$A$7;0));"FALTA TARIFA";"ok").

Junto al total: =SI(CONTAR.SI(F2:F4;"FALTA TARIFA")>0;"REVISAR: tarifa ausente";"Tarifas

completas").

CLAVE

Con Formato › Formato condicional puedes ponerlo en rojo, pero el aviso debe ser texto, no solo color.

1

Qué se vio

I-01, 19/10/2027: sin tarifa

de A003.

2

Qué efecto tenía

39,07 € menos con el

cero silencioso.

3

Cómo se corrigió

Alta de A003 a 11,33 en

Tarifas.

4

Qué lo confirma

F4 en «ok» y total de

138,16.

🙋  PREGUNTA  PARA  LA  CLASE

Isabela envuelve la columna de precios en SI.ERROR(…;0)

porque dos #N/A le parecían feos. El total queda en 184,40
€. ¿Qué ha pasado?

✔ Lectura y debate «El presupuesto que salió barato»: los herrajes costaban 622,70 €; 622,70 − 184,40 =
438,30 €, que salieron del margen. «Los supuestos fuera de la fórmula, y los huecos a la vista».

A Nada: el total es correcto

B Bisagras y tiradores no estaban en la lista y la hoja los contó a cero: faltaban 438,30 €

C SI.ERROR ha redondeado los precios

6 .5  ·  LA  NORMA  DICE

Plazo de pago sin pacto

⚖️
Si el contrato no fija plazo, el pago es a treinta días

naturales desde la recepción de las mercancías o la

prestación del servicio, aunque la factura llegue

antes; por pacto puede ampliarse hasta sesenta.

Ley 3/2004, de 29 de diciembre, por la que se establecen medidas
de lucha contra la morosidad en las operaciones comerciales, art. 4
· Consulta la versión vigente en el BOE

SUJETO OBLIGADO

Las partes de una operación comercial.

Si hay plazo pactado, rige el contrato.

NO OBLIGA A…

…a contar el vencimiento desde la

factura: en el modelo, H4 = 30 días

desde la factura es el plazo pactado con

el cliente, un supuesto que se cambia

en su celda.

CLAVE

Facturado no es cobrado: una factura emitida es un derecho de cobro, no dinero en el banco.

6 .5  ·  HOJA  CONTROL

El estado de la factura de P002
P002 (242,00 €), facturado el lunes 04/10/2027, vence el miércoles 03/11/2027. En Control, H4

toma el plazo de Presupuesto (=Presupuesto!H4, 30 días) y H5 es la fecha de control:

martes 19/10/2027, tecleada.

CELDA CONCEPTO FÓRMULA RESULTADO

B2 Fecha de factura 04/10/2027 04/10/2027

D2 Vencimiento =B2+$H$4 03/11/2027

G2 Días hasta el

vencimiento

=D2-$H$5 15

F2 Estado =SI(E2<>"";"Cobrada";SI($H$5>D2;"Vencida";"Pendiente de

cobro"))

Pendiente de

cobro

E2, la fecha de cobro, está vacía mientras no se cobra. Con H5 en el jueves 04/11/2027, G2 pasa a −1 y el estado a Vencida; con una fecha en E2, a Cobrada. HOY()
cambia cada día y la hoja daría mañana otro estado.

CLAVE

El estado depende de un dato registrado (la fecha de cobro), no de que la factura exista.

6 .5  ·  VENCIMIENTOS ,  PENDIENTES  Y  VALIDACIONES

Lo que la hoja controla

📆
Días no laborables

Factura del viernes 01/10/2027 a 30 días: domingo 31/10/2027.

Con ese vencimiento en D6 de Control, =DIA.LAB(D6-

1;1;Festivos), con el lunes 01/11/2027 en Festivos

(Control!K2:K3), planifica el martes 02/11/2027.

🧾
Los pendientes

=SUMAR.SI(B8:B11;"Pendiente";C8:C11) suma P001 (138,16) y

P003 (200,00): 338,16 €. La suma de los cuatro, 640,66 €,

incluye todos los estados.

🚫
Validación de cantidad

C2:C4, fórmula personalizada

=Y(ESNUMERO(C2);C2>=1;C2=ENTERO(C2)) y rechazar la

entrada: 0 o 2,5 no entran.

📋
Validación del descuento

H2, menú desplegable de un intervalo con 0 %, 5 %, 8 % y 10

%, escritos en la hoja Tarifas.

CLAVE

Una validación de datos impide o avisa de entradas imposibles: pruébala con un dato bueno y con uno malo.

6 .5  ·  ALTERAR  UN  SUPUESTO

PR-002 con cuatro descuentos
Taller Brezo, 200,00 € de base (A004 × 20 a 4,50 y A005 × 5 a 22,00). Solo se cambia H2.

DESCUENTO DESCUENTO (€) BASE (€) IVA (€) TOTAL (€)

0 % 0,00 200,00 42,00 242,00

5 % 10,00 190,00 39,90 229,90

8 % 16,00 184,00 38,64 222,64

10 % 20,00 180,00 37,80 217,80

Se anota el escenario probado antes de volver al valor de partida y se deja una tabla de escenarios en Control. A mano, el 8 %: 200 × 0,08 = 16; 184 × 0,21 =
38,64.

CLAVE

Cambiar un supuesto prueba que el modelo sirve.

6 .5  ·  COMPROBAR  POR  UN  CAMINO  INDEPENDIENTE

Control con tolerancia

=REDONDEAR(SUMAPRODUCTO(C2:C4;D2:D4)*(1-H2)*(1+H3);2)

PR-001 con 10 unidades modelo 138,16 control 138,16

PR-001 con 11 unidades modelo 142,19 control 142,18

Por qué el modelo redondea cada paso (6,18; 24,68)

el control, solo al
final: 117,5055 ×
1,21 = 142,181655

Celdas J1: =SUMAPRODUCTO(C2:C4;D2:D4) · J2: el total de control J1, J2 y J3

Tolerancia J3: =SI(ABS(E10-J2)<=0,02;"coincide";"REVISAR")

0,50 € de
diferencia sería

un error

Comprobar un total con la misma fórmula que lo calculó no comprueba nada. Se anota en la hoja quién hizo la comprobación y cuándo.

CLAVE

El céntimo de diferencia es esperable: por eso el control lleva una tolerancia anotada.

CASO  RESUELTO  ·  PASO  A  PASO

PR-002 a Taller Brezo con tres escenarios

🧮
Carolina, con Teodoro

Evaristo quiere ver el 5 % y el

10 %. Factura del lunes

04/10/2027, plazo de 30 días.

Caso ficticio.

1 Los datos de partida. A004 × 20 a 4,50 y A005 × 5 a 22,00 por BUSCARV. En

Presupuesto, H2 0 %, H3 21 % y H4 30 días; en Control, H4 repite el plazo y

H5 es la fecha de control, martes 19/10/2027.

2 Las líneas. E2 =REDONDEAR(C2*D2;2): 90,00; E3: 110,00; subtotal 200,00 €.

3 Los escenarios. Solo cambia H2: 242,00 € (0 %), 229,90 € (5 %) y 217,80 € (10

%).

4 El cobro. D2 =B2+$H$4: miércoles 03/11/2027; faltan 15 días: Pendiente de

cobro.

5 El control independiente. =REDONDEAR(SUMAPRODUCTO(C2:C3;D2:D3)*

(1-H2)*(1+H3);2) coincide en los tres. A mano: 90 + 110 = 200; 200 × 1,21 = 242.

✔ 242,00 €, 229,90 € y 217,80 €, con el control coincidente; la factura de
242,00 € vence el miércoles 03/11/2027 y hoy está pendiente de cobro.

OJO  PROFES IONAL

Revisión de PR-001 marcada como APROBADO
Documento ficticio, preparado por Carolina, con 11 unidades de A001. Fuente: (sin

identificar) · Comprobación: (no registrada). Contiene cinco errores.

N.º CASILLA CONTENIDO

1 E9 =REDONDEAR(E8*0,21;2): el IVA dentro de la fórmula y H3 sin usar

2 D4 =SI.ERROR(BUSCARV(…);0): sin tarifa de A003, total 99,09 sin aviso

3 E7, E9 y E10 Sin REDONDEAR, solo formato de dos decimales: muestra 142,18

4 F2 «Cobrada» para P002 solo porque tiene fecha de factura

5 J3 Control de totales «igual que E10» (=E10), sin comprobación independiente ni tolerancia

CLAVE

Correcciones: =REDONDEAR(E8*$H$3;2); quitar SI.ERROR y avisar con FALTA TARIFA; REDONDEAR

descuento (6,18) e IVA (24,68): 142,19; Pendiente de cobro hasta que conste el cobro; control con

SUMAPRODUCTO y tolerancia. Y en la cabecera, fuente, comprobación y fecha.

APRENDE  CON  LA  IA

La respuesta que recibió Carolina
Carolina preguntó a un asistente de IA por las tarifas ausentes, el IVA y el cobro. Pulsa cada

tarjeta.

«Envuelve todas

las fórmulas en
SI.ERROR(…;0):

nunca verás un
aviso»

BULO

Sin la tarifa de A003,

el total sale 99,09 €

en lugar de 138,16 €.

El hueco debe

quedar a la vista:

columna FALTA

TARIFA y alerta.

«Para el IVA, escribe

=E8*1,21 en cada
celda»

BULO

El 21 % es un supuesto: va

en H3 y las fórmulas

apuntan a él,

=REDONDEAR(E8*$H$3;2).

«Si la factura tiene

fecha, el cobro es
seguro»

BULO

Facturado no es

cobrado: P002 vence

el 03/11/2027 y está

pendiente hasta que
conste la fecha de

cobro.

«Con 11 unidades

de A001 el total es
142,18 €»

BULO

142,18 es el resultado

sin redondear cada

paso. El modelo

correcto da 142,19 €
(descuento 6,18; IVA

24,68).

TU  CARPETA  OPERAT IVA  ·  PASO  6

Modelo comercial
Guarda en 00 Carpeta operativa el archivo UD06_P06_modelo_comercial_v01:

CLAVE

Comprobación a mano de la práctica: A001 = 35,00 € y A003 = 33,99 €.

1

Cinco hojas

Origen_tarifas, Registro, Tarifas,

Presupuesto y Control, con los

escenarios y la plantilla.

2

Tres cifras de control

Total de 138,16 €, el de 11 unidades,

142,19 €, y los pendientes, 338,16 €.

3

Lo que debe verse

Las fórmulas de D2, E7 y E9, la

incidencia de tarifa ausente y la

comprobación a mano de dos

líneas, con quién y cuándo.

REPASO

Ideas clave

✓ Se importa con Archivo › Importar › Subir, separador

punto y coma y conversión desactivada: CONTARA da 6

y CONTAR, 0. Origen_tarifas no se toca.

✓ Cada columna se normaliza con una regla escrita:

ESPACIOS, VALOR(SUSTITUIR(…)) y FECHA(EXTRAE(…));

después, Pegado especial › Solo valores.

✓ Las fechas son números de días (04/10/2027 = 46664);

una celda vacía no es un cero.
✓ Un modelo separa entradas, cálculos y resultados, con

los supuestos (descuento, IVA, plazo) en celdas propias.

✓ PR-001: 120,19 − 6,01 = 114,18 + 23,98 = 138,16 €; con 11

unidades de A001, 142,19 €.
✓ BUSCARV con FALSO sigue al código aunque se

ordene Tarifas; ÍNDICE y COINCIDIR miran a la

izquierda.

✓ Una tarifa ausente se ve: #N/A, FALTA TARIFA y alerta.

SI.ERROR(…;0) la convierte en un total falso de 99,09 €.
✓ Facturado no es cobrado: vencimiento =B2+$H$4

(03/11/2027) y estado según la fecha de cobro; la fecha

de control se teclea.

✓ Pendientes con SUMAR.SI (338,16 €), validaciones

probadas, escenarios en H2 y control independiente

con tolerancia.

6

🚀
¡Ahora te toca a ti!
Repasa, practica y comprueba lo que has aprendido.

📝
Resumen

✅
Test

👆
Interactivos

Libro, presentaciones, resúmenes y tests gratis en josesande.com

